
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 22.4.23 which is part of the NBIOT test suite in the ‘iwd-TTCN3-B2017-06_D17wk37’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Yongsheng MA

mayongsheng@starpointcomm.com

Tel. +8615117939410

1.1 Verification Test Summary

Test Case:
TC_22_4_23
ATS Version:
iwd-TTCN3-B2017-06_D17wk37
System Simulator used:
Starpoint SP8630

UE used:
Hisilicon Hi2110
Verification Status:
PASS
2. Corrections required

2.1 Change 1 – Correction to template ‘cs_RLF_TimersAndConstants_Setup’

	Template name
	cs_RLF_TimersAndConstants_Setup

	Reason for change
	The Table 22.4.23.3.3-2 of associated Prose CR changes t310_r13 to ms8000, so parameter p_T310 needs to pass in the template cs_RLF_TimersAndConstants_Setup.

	Summary of change
	Add parameter p_T310 to template cs_RLF_TimersAndConstants_Setup.
Associated Prose CR will be submitted in next RAN5 meeting.

	TTCN module
	NBIOT_RRC_Templates.ttcn

	MCC160 Comment
	Accepted


Before change

  template (value) RLF_TimersAndConstants_NB_r13 cs_RLF_TimersAndConstants_Setup(RLF_TimersAndConstants_NB_r13.setup.t301_r13 p_T301) :=

  { /* @status    APPROVED (NBIOT) */

    setup := {

      t301_r13  := p_T301,

      t310_r13  := ms1000,

      n310_r13  := n1,

      t311_r13  := ms1000,

      n311_r13  := n1,

      t311_v1350 := omit                          /* @sic R5s170597 BASELINE MOVING 2017 sic@ */

    }

  };
After change

  template (value) RLF_TimersAndConstants_NB_r13 cs_RLF_TimersAndConstants_Setup(RLF_TimersAndConstants_NB_r13.setup.t301_r13 p_T301,

                                                                                 RLF_TimersAndConstants_NB_r13.setup.t310_r13 p_T310 := ms1000) :=

  { /* @status    APPROVED (NBIOT) */

    setup := {

      t301_r13  := p_T301,

      t310_r13  := p_T310,

      n310_r13  := n1,

      t311_r13  := ms1000,

      n311_r13  := n1,

      t311_v1350 := omit                          /* @sic R5s170597 BASELINE MOVING 2017 sic@ */

    }

  };
2.2 Change 2 – Correction to template ‘cs_RLF_TimersAndConstants_Setup_CP’

	Template name
	cs_RLF_TimersAndConstants_Setup_CP

	Reason for change
	The Table 22.4.23.3.3-2 of associated Prose CR changes t310_r13 to ms8000, and change 1 adds parameter p_T310 to the template cs_RLF_TimersAndConstants_Setup.

	Summary of change
	Change t310 to ms8000 according to associated Prose CR.
Associated Prose CR will be submitted in next RAN5 meeting.

	TTCN module
	NBIOT_RRC_Templates.ttcn

	MCC160 Comment
	Accepted


Before change

  template (value) RLF_TimersAndConstants_NB_r13 cs_RLF_TimersAndConstants_Setup_CP := cs_RLF_TimersAndConstants_Setup(ms10000);
After change

  template (value) RLF_TimersAndConstants_NB_r13 cs_RLF_TimersAndConstants_Setup_CP := cs_RLF_TimersAndConstants_Setup(ms10000, ms8000);
2.3 Change 3 – Correction to function ‘function f_TC_22_4_23_NBIOT()’

	Function name
	f_TC_22_4_23_NBIOT()

	Reason for change
	1. The associated Prose CR only used cell1 for testing and deleted cell2, so remove the ttcn codes related cell2.
2. In order that SERVICE ACCEPT message has been received successfully by UE befor cell becomes off, add 1s delay after steps 1/1G.
3. Steps 3-7 check UE will initiate a TAU procedure and transmit a RRCConnectionRequest-NB message with 8s. But it is possible a conformant UE will spend more time to select new cell and establish a RRC connection after t310-r13(1s) expired. This test procedure is not appropriate. The following procedure is better to achieve its Test Purpose 1: Using Require UE Capability Information to Check UE will keep in RRC_CONNECTED when radio link has been recovered and t310 is running. Void steps 2 and 3-7.
4. According to new TP2 of associated Prose CR, Change/add steps 8-14 to check UE will use t310 SystemInformationBlockType2-NB and initiate a TAU procedure when t310 expired.

	Summary of change
	1. Delete the ttcn codes related cell2 and v_CellPowerList_AtT1/T2, use function f_NBIOT_SetCellPower() to change cell power for cell1.
2. Add 1s delay after steps 1/1G.
3. Add new steps 1A-1G to check whether UE use the t310 value configured in rlf-TimersAndConstants-r13. Void steps 2 and 3-7.

4. Change/add steps 8-14 to check UE will use t310 SystemInformationBlockType2-NB and initiate a TAU procedure when t310 expired.
Associated Prose CR will be submitted in next RAN5 meeting.

	TTCN module
	NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	1, 3, 4 Accepted
2. Accepted in principle but implemented 500ms.


Before change

  function f_TC_22_4_23_NBIOT() runs on NBIOT_PTC

  {//NB-IoT / Radio link failure / T310 expiry / Dedicated RLF timer (CP CIoT)

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT2;
    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var template (value) DL_CCCH_Message_NB v_RrcConnSetup;

    timer t_Wait;
    v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, -85)

    };
    v_CellPowerList_AtT2 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };
    f_NBIOT_Init(c1);

    //Set the value of T311 timer in SIB2 for Ncell 1, Ncell 2 according to 22.4.23.3.3-1

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB2.ue_TimersAndConstants_r13.t310_r13 := ms8000;
    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell2);

    v_SIB2.ue_TimersAndConstants_r13.t310_r13 := ms8000;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell2, v_SIB2);
    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell2, CONTROL_PLANE);
    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE3_NB_IDLEUPDATED);

    f_NBIOT_TestBody_Set(true);

    // Setup the RRC Connection Setup message to be sent in step 1

    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup();

    v_RrcConnSetup.message_.c1.rrcConnectionSetup_r13.criticalExtensions.c1.rrcConnectionSetup_r13.radioResourceConfigDedicated_r13.rlf_TimersAndConstants_r13 := cs_RLF_TimersAndConstants_Setup_CP;

    f_NBIOT_SS_ConfigRachProcedureMsg4 (nbiot_Cell1, v_RrcConnSetup);

    //@siclog "Step 1" siclog@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 2" siclog@

    //The SS changes Ncell1 and Ncell2 levels according to row "T1" in Table 22.4.23.3.2-1.
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);
    // Setup the RRC Connection Setup message to be sent in step 4

    v_RrcConnSetup.message_.c1.rrcConnectionSetup_r13.criticalExtensions.c1.rrcConnectionSetup_r13.radioResourceConfigDedicated_r13.rlf_TimersAndConstants_r13 := cs_RLF_TimersAndConstants_Release;

    f_NBIOT_SS_ConfigRachProcedureMsg4 (nbiot_Cell1, v_RrcConnSetup);
    //@siclog "Step 3-7" siclog@

    //Check: Does the UE perform steps 1 to 5 of the generic test procedure in TS 36.508 subclause 8.1.5A.5 on Ncell 2 within 8s.

    t_Wait.start(8.0);

    alt {

    //NOTE: The UE performs a TAU procedure and enters into E-UTRA RRC_CONNECTED state

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell2, cr_RRCConnectionRequest_NB(mo_Signalling)))){

        f_NBIOT_TrackingAreaUpdate_Step1_5(nbiot_Cell2, CONTROL_PLANE);

        t_Wait.stop;

        f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 3-7");

        }

      [] t_Wait.timeout{

        f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 3-7");

      }

    }
    //@siclog "Step 8" siclog@

    //The SS changes Ncell1 and Ncell2 levels according to row "T2" in Table 22.4.23.3.2-1.
    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT2);
    //@siclog "Step 9" siclog@
    //Check: Check: Does the UE initiate an RRCConnectionRequest-NB on Ncell 1 within 8s.

    if (f_NBIOT_RRC_RRCConnectionRequest_Check(nbiot_Cell1, f_NBIOT_SetTimerToleranceMax(rrcTimer, 8.0)) ){

      f_NBIOT_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 9");

    }
    //@siclog "Step 10" siclog@

    //Check: Does the UE perform steps 1 to 5 of the generic test procedure in TS 36.508 subclause 8.1.5A.5 on Ncell1.

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_NBIOT_TrackingAreaUpdate(nbiot_Cell11, CONTROL_PLANE);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 10");
    f_NBIOT_TestBody_Set(false);

    // Now configure Msg4 back to the default RRC Conn Setup for the postamble

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());
    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }
After change

  function f_TC_22_4_23_NBIOT() runs on NBIOT_PTC

  {//NB-IoT / Radio link failure / T310 expiry / Dedicated RLF timer (CP CIoT)

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var template (value) DL_CCCH_Message_NB v_RrcConnSetup;

    var NAS_MSG_Indication_Type v_NasInd;
    var GutiParameters_Type v_GutiParams;
    f_NBIOT_Init(c1);

    //Set the value of T311 timer in SIB2 for Ncell 1, Ncell 2 according to 22.4.23.3.3-1

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);
    v_SIB2.ue_TimersAndConstants_r13.n310_r13 := n1;
    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE3_NB_IDLEUPDATED);

    v_GutiParams:= f_NBIOT_CellInfo_GetGuti(nbiot_Cell1);
    f_NBIOT_TestBody_Set(true);

    // Setup the RRC Connection Setup message to be sent in step 1

    v_RrcConnSetup := f_NBIOT_508_RRCConnectionSetup();

    v_RrcConnSetup.message_.c1.rrcConnectionSetup_r13.criticalExtensions.c1.rrcConnectionSetup_r13.radioResourceConfigDedicated_r13.rlf_TimersAndConstants_r13 := cs_RLF_TimersAndConstants_Setup_CP;

    f_NBIOT_SS_ConfigRachProcedureMsg4 (nbiot_Cell1, v_RrcConnSetup);

    //@siclog "Step 1" siclog@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);
    f_Delay(1.0);//Wait for SS transmitting DL NAS massage SERVICE ACCEPT
    //@siclog "Step 1A" siclog@

f_NBIOT_SetCellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE);
    //@siclog "Step 1B" siclog@
    f_Delay(5.0);
    //@siclog "Step 1C" siclog@
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
    //@siclog "Step 1D" siclog@
    SRB.send(cas_NB_SRB_RrcPdu_REQ(nbiot_Cell1, tsc_SRB1bis, cs_TimingInfo_Now, cs_508_UeCapabilityEnquiry_NB(tsc_RRC_TI_Def)));
    //@siclog "Step 1E" siclog@
    SRB.receive(car_NB_SRB_RrcPdu_IND(nbiot_Cell1, tsc_SRB1bis, cr_UeCapabilityInformation_NB_Any(tsc_RRC_TI_Def)));
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 1E");
    //@siclog "Step 1F" siclog@
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);
    //@siclog "Step 1G" siclog@
    //Now configure Msg4 back to the default RRC Conn Setup
    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());
f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);
    f_Delay(1.0);//Wait for SS transmitting DL NAS massage SERVICE ACCEPT
    //@siclog "Step 2" siclog@

    //Void.
    //@siclog "Step 3-7" siclog@

    //Void.
    //@siclog "Step 8" siclog@

    //The SS changes Ncell1 and Ncell2 levels according to row "T2" in Table 22.4.23.3.2-1.

    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE); 
    f_NBIOT_SS_SRBs_DRBs_Reset (nbiot_Cell1, CONTROL_PLANE);
    //@siclog "Step 8A" siclog@
f_Delay(3.0);
    //@siclog "Step 8B" siclog@
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
    //@siclog "Step 9-11" siclog@
    v_NasInd := f_NBIOT_TrackingAreaUpdate_Step1_3(nbiot_Cell1, ?);
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    //@siclog "Step 12" siclog@
    //The SS responds with TRACKING AREA UPDATE ACCEPT message.
    //@siclog "Step 13" siclog@
    //The UE transmits a TRACKING AREA UPDATE COMPLETE message.
    f_NBIOT_TrackingAreaUpdate_Accept_Complete(nbiot_Cell1,

                                               CONTROL_PLANE,

                                               tsc_SRB1bis,

                                               v_GutiParams,

                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus,

                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType);
    //@siclog "Step 14" siclog@
    //The SS transmits an RRCConnectionRelease-NB message.
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);
    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }
3. Branches executed

None.

4. Execution Log Files

4.1 Hisilicon Hi2110

The Hisilicon Hi2110 passed this test case on Starpoint SP8630 Integration Test System in LTE FDD band 8. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
TC_22_4_23_Hi2110_PASS.spm

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT file name
TC_22_4_23.sppar

5. References

	[1]
	R5s170782: Supporting information for agreement of NB-IoT test case 22.4.23. This archive comprises html and text format execution log files and PICS/PIXIT settings file respectively.


